	[image: image1.png]T MANEMZTHMIO

MAKEAON]AZ
OIKONOMIKAN KAI KOINONIKON EMIETHMON




ΤΜΗΜΑ ΕΦΑΡΜΟΣΜΕΝΗΣ ΠΛΗΡΟΦΟΡΙΚΗΣ

ΕΡΓΑΣΤΗΡΙΟ ΠΑΡΑΛΛΗΛΗΣ ΚΑΙ ΚΑΤΑΝΕΜΗΜΕΝΗΣ ΕΠΕΞΕΡΓΑΣΙΑΣ
	
	Μάθημα
: Παράλληλη Επεξεργασία

Φύλλο Ασκήσεων
: 6 – Σύγχρονος Παραλληλισμός
Ονοματεπώνυμο
: _______________________________

Α.Μ.
: __________________

Email
: __________________

Εξάμηνο
: _______


Παράλληλη Επεξεργασία

Προγραμματιστικές Ασκήσεις

ΠΡΟΓΡΑΜΜΑΤΙΣΤΙΚΕΣ ΑΣΚΗΣΕΙΣ
Άσκηση 1η 

Το παρακάτω πρόγραμμα επιλύει τον αλγόριθμο χαλάρωσης του Jacobi σε ένα παράλληλο σύστημα διαμοιραζόμενης μνήμης με τη χρήση φράγματος (Barrier), που έχει υλοποιηθεί με την τεχνική του κλειδώματος (Spinlock). Εκτελέστε το πρόγραμμα στην MPWin. Στη συνέχεια δημιουργήστε τα γραφήματα Process Activity Chart και  Memory Access Pattern για τις μεταβλητές Count, Arrival και Departure ως προς τις θέσεις μνήμης και τις παράλληλες διεργασίες (Memory Location, Process View). Εντοπίστε και σημειώστε πάνω στα γραφήματα για την εκτέλεση της πρώτης επανάληψης της διεργασίας 1 φάσεις υπολογισμού κι ενημέρωσης των δομών δεδομένων καθώς και τις φάσεις άφιξης κι αναχώρησης μέσα στο φράγμα. Σε ποιον τύπο προβολής (Memory Location ή Process View) του γραφήματος Process Memory Access Pattern είναι πιο ευδιάκριτες οι φάσεις του προγράμματος; (Σημείωση: Λάβετε υπόψη ότι δεν υπάρχει ακριβής αντιστοιχία του παράλληλου χρόνου εκτέλεσης στα δύο γραφήματα, κι αυτό γιατί το Process Activity Chart παρουσιάζει μια προσεγγιστική εικόνα των καταστάσεων των παράλληλων διεργασιών με βήμα 10 χρονικών μονάδων.)
Program JacobiBarrier;

Const 

  n = 4;

  numiter =5;

Var 

  i, j : Integer;

  a, b : Array [0..n+1,0..n+1] of Real;  

  count : Integer;

  Arrival, Departure : Spinlock;

Procedure Barrier;

begin

  (*Φάση Άφιξης στο φράγμα*)

  Lock(Arrival);

  count := count + 1;

  If count < n then Unlock(Arrival)  (*Συνέχιση της Φάσης Άφιξης*)

  else Unlock(Departure);            (*Τέλος της Φάσης Άφιξης*)

  (*Φάση Αναχώρησης από το φράγμα*)

  Lock(Departure);

  count := count - 1;

  If count > 0 then Unlock(Departure) (*Συνέχιση της Φάσης Αναχώρησης*)

  else Unlock(Arrival);               (*Τέλος της Φάσης Αναχώρησης*)

end;

Begin
  (*Αρχικοποίηση της μεταβλητής count και των κλειδωμάτων*)

  count := 0;

  Unlock(Arrival);

  Lock(Departure);

  (*Αρχικοποίηση των τιμών για τον πίνακα Α*)

  For i:=1 to n do

    For j:=1 to n do

      a[i,j]:=0;

  For i:=0 to n+1 do begin

    a[i,0]:=15;a[i,n+1]:=15;

    a[0,i]:=15;a[n+1,i]:=15;

  end;

  B := A;

  Forall i := 1 to n do 

  (*Φάση υπολογισμού*)

    Var j, k : Integer;

    begin 

      For k := 1 to numiter do begin

        For j := 1 to n do 

          B[i,j] := (A[i-1,j]+A[i+1,j]+A[i,j-1]+A[i,j+1]) / 4;

        Barrier;

        (*Φάση ενημέρωσης των δομών δεδομένων*)

        A[i] := B[i];

        Barrier;

      end;

    end;

  For i:=0 to n+1 do begin

    For j:=0 to n+1 do 

      write(A[i,j]:5:2,' ');

    writeln;

  end;

End.

Άσκηση 2η 

Το παρακάτω πρόγραμμα επιλύει τον αλγόριθμο χαλάρωσης του Jacobi σε ένα παράλληλο σύστημα διαμοιραζόμενης μνήμης με τη χρήση φράγματος (Barrier), που έχει υλοποιηθεί με την τεχνική του δυαδικου δέντρου (Tournament technique) και μεταβλητές κανάλια (Channels). Εκτελέστε το πρόγραμμα στην MPWin. Στη συνέχεια δημιουργήστε τα γραφήματα Process Activity Chart, Memory Access Pattern για την μεταβλητή Synchan ως προς τις θέσεις μνήμης και τις παράλληλες διεργασίες (Memory Location, Process View) και Channel Communication Chart. Εντοπίστε και σημειώστε πάνω στα γραφήματα για την εκτέλεση της πρώτης επανάληψης της διεργασίας 1 τις φάσεις υπολογισμού κι ενημέρωσης των δομών δεδομένων καθώς και τις φάσεις αποστολής και λήψης μηνυμάτων μεταξύ των διεργασιών μέσα στο φράγμα. Σε ποιον τύπο προβολής (Memory Location ή Process View) του γραφήματος Process Memory Access Pattern είναι πιο ευδιάκριτες οι φάσεις του προγράμματος;
Program JacobiBarrier;

Const 

  n = 4; (*Δύναμη του 2*)

  numiter =5;

Var 

  i, j : Integer;

  a, b : Array [0..n+1,0..n+1] of Real;

  synchan: Array [0..n-1] of Channel of Integer;

Procedure Barrier(me: Integer);

var

  position, dummy: Integer;

begin

  position := 1;

  While (me DIV position MOD 2 = 0) AND (position < n) do begin

    dummy:=synchan[me];  (*Λήψη μηνύματος από τον αντίπαλο που έχει χάσει*)

    position := position * 2;

  end;

  If me <> 0 then BEGIN

    synchan[me-position] := 1;  (*Διάδοση μηνύματος στη διεργασία νικητή*)

    dummy := synchan[me];       (*Αναμονή για απάντηση*)

  end;

  While position > 1 do begin

    position := position DIV 2;

    synchan[me+position] := 1;  (*Αποστολή μηνύματος στη διεργασία που είχε χάσει προηγουμένως*)

  end; 

end;

Begin

 (*Αρχικοποίηση των τιμών για τον πίνακα Α*)

  For i:=1 to n do

    For j:=1 to n do

      a[i,j]:=0;

  For i:=0 to n+1 do begin

    a[i,0]:=15;a[i,n+1]:=15;

    a[0,i]:=15;a[n+1,i]:=15;

  end;

  B := A;

  Forall i := 1 to n do (@i-1)

    (*Φάση υπολογισμού*)

    Var

      j, k : Integer;

    begin 

      For k := 1 to numiter do begin

        For j := 1 to n do 

          B[i,j] := (A[i-1,j]+A[i+1,j]+A[i,j-1]+A[i,j+1]) / 4;

        Barrier(i-1);

        (*Φάση ενημέρωσης των δομών δεδομένων*)

        A[i] := B[i];

        Barrier(i-1);

      end;

    end;

  For i:=0 to n+1 do begin

    For j:=0 to n+1 do write(A[i,j]:5:2,' ');

    writeln;

  end;

End.

Άσκηση 3η 

Το παρακάτω πρόγραμμα επιλύει τον αλγόριθμο χαλάρωσης του Jacobi σε ένα παράλληλο σύστημα διαμοιραζόμενης μνήμης με έλεγχο σύγκλισης. Για τον συγχρονισμό των παράλληλων διεργασιών χρησιμοποιήθηκαν μία διαδικασία φράγματος (Barrier) και μία συνάρτηση συλλογής για τον έλεγχο της σύγκλισης, που έχουν υλοποιηθεί με την τεχνική του κλειδώματος (Spinlock). Εκτελέστε το πρόγραμμα στην MPWin. Στη συνέχεια δημιουργήστε τα γραφήματα Process Activity Chart και Memory Access Pattern για τις μεταβλητές Count, Arrival, Departure και Globaldone αρχικά ως προς τις θέσεις μνήμης κι έπειτα ως προς τις παράλληλες διεργασίες (Memory Location, Process View). Εντοπίστε και σημειώστε πάνω στα γραφήματα για την εκτέλεση της πρώτης επανάληψης της διεργασίας 1 τις φάσεις υπολογισμού κι ενημέρωσης των δομών δεδομένων καθώς και τις φάσεις άφιξης κι αναχώρησης μέσα στη διαδικασία φράγματος Barrier και μέσα στην συνάρτηση συλλογής Aggregate. Σε ποιον τύπο προβολής (Memory Location ή Process View) του γραφήματος Process Memory Access Pattern είναι πιο ευδιάκριτες οι φάσεις του προγράμματος;
Program Jacobi_Convergence;

Const 

  tolerance = .1;

  n = 4;

Var  

  A, B : Array [0..n+1,0..n+1] of Real;

  i,j : integer;

  count : Integer;

  Arrival, Departure : Spinlock;

  globaldone : Boolean;

Procedure Barrier;

begin

  (*Φάση Άφιξης στο φράγμα*)

  Lock(Arrival);

  count := count + 1;

  If count < n then Unlock(Arrival)  (*Συνέχιση της Φάσης Άφιξης*)

  else Unlock(Departure);            (*Τέλος της Φάσης Άφιξης*)

  (*Φάση Αναχώρησης από το φράγμα*)

  Lock(Departure);

  count := count - 1;

  If count > 0 then Unlock(Departure) (*Συνέχιση της Φάσης Αναχώρησης*)

  else Unlock(Arrival);               (*Τέλος της Φάσης Αναχώρησης*)

end;

Function Aggregate(mydone : Boolean) : Boolean;

BEGIN

  (*Φάση Άφιξης - Άθροιση των διεργασιών καθώς εισέρχονται στο φράγμα*)

  Lock(Arrival);

  count := count+1;

  globaldone := globaldone AND mydone;  (*Βήμα σύγκλισης*)

  If count < n then Unlock(Arrival)   (*Συνέχιση της Φάσης Άφιξης*)

  else Unlock(Departure); (*Τερματισμός της Φάσης Άφιξης*)

  (*Φάση Αναχώρησης - Άθροιση των διεργασιών καθώς εξέρχονται από το φράγμα*)

  Lock(Departure);

  count := count - 1;

  Aggregate:=globaldone;(*Επιστροφή της σημαίας flag στη διεργασία*)

  If count > 0 then Unlock(Departure)  (*Συνέχιση της Φάσης Αναχώρησης*)

  else begin

    globaldone := TRUE; (*Αλλαγή της τιμής της μεταβλητής για την επόμενη επανάληψη*)

    Unlock(Arrival);  (*Τερματισμός της Φάσης Αναχώρησης*)

  end;

end;

Begin (*Κυρίως πρόγραμμα*)

  (*Αρχικοποίηση της μεταβλητής count και των κλειδωμάτων*)

  count := 0;

  Unlock(Arrival);

  Lock(Departure);

  globaldone := TRUE;  (*Αρχικοποίηση της τοπικής σημαίας*)

 (*Αρχικοποίηση των τιμών για τον πίνακα Α*)

  For i:=1 to n do

    For j:=1 to n do

      a[i,j]:=0;

  For i:=0 to n+1 do begin

    a[i,0]:=15;a[i,n+1]:=15;

    a[0,i]:=15;a[n+1,i]:=15;

  end;

  B := A;

  Forall i := 1 to n do  (*Δημιουργία των διεργασιών*)

    (*Φάση υπολογισμού*)

    Var 

      j: Integer;

      change, maxchange : Real;

      done : Boolean;

    begin

      Repeat

        maxchange := 0;

        For j := 1 to n do begin (*Υπολογισμός νέων τιμών για κάθε στοιχείο της γραμμής*)

          B[i,j] := (A[i-1,j]+A[i+1,j]+A[i,j-1]+A[i,j+1])/4;

          change := ABS(B[i,j]-A[i,j]);

          If change > maxchange then maxchange := change; 

        end;

        Barrier;

        (*Φάση ενημέρωσης δομών δεδομένων*)

        A[i] := B[i];

        done := Aggregate(maxchange<tolerance);

      Until done; (*Επανάληψη μέχρι τον καθορισμό του τοπικού τερματισμού*)

    end;

  For i:=0 to n+1 do begin

    For j:=0 to n+1 do write(A[i,j]:6:2);

    writeln;

  end;

End.

Άσκηση 4η 

Τροποποιήστε το πρόγραμμα της 3ης άσκησης με έλεγχο σύγκλισης, έτσι ώστε να υλοποιήστε την διαδικασία του φράγματος Barrier και την συνάρτηση συλλογής Aggregate με την χρήση μεταβλητών καναλιών (Channels) αντί των μεταβλητών κλειδώματος (Spinlocks). Εκτελέστε το πρόγραμμα στην MPWin. Στη συνέχεια δημιουργήστε τα γραφήματα Process Activity Chart,  Memory Access Pattern για τις μεταβλητές τύπου Channel και Globaldone ως προς τις θέσεις μνήμης και τις παράλληλες διεργασίες (Memory Location, Process View). Εντοπίστε και σημειώστε πάνω στα γραφήματα για την εκτέλεση της πρώτης επανάληψης της διεργασίας 1 τις φάσεις υπολογισμού κι ενημέρωσης των δομών δεδομένων καθώς και τις φάσεις αποστολής και λήψης μηνυμάτων μεταξύ των παράλληλων διεργασιών μέσα στη διαδικασία φράγματος Barrier και μέσα στην συνάρτηση συλλογής Aggregate. Στη συνέχεια συγκρίνετε τη συμπεριφορά των προγραμμάτων 3 και 4 ως προς τo γράφημα Process Activity Chart.
Άσκηση 5η 

Το παρακάτω πρόγραμμα χρησιμοποιεί μία τεχνική τοπικού φράγματος για τον συγχρονισμό των παράλληλων διεργασιών. Η τροποποιημένη διαδικασία LocalBarrier θα οδηγήσει σε αδιέξοδο. Εκτελέστε το πρόγραμμα στην ΜPWin κι αναλύστε για ποιο λόγο δημιουργείται το αδιέξοδο. Χρησιμοποιήσατε κάποια γραφήματα για να διαπιστώσετε και να εξηγήσετε την εμφάνιση του αδιέξοδου; Αν ναι, ποια γραφήματα χρησιμοποιήσατε και πως αυτά σας βοήθησαν;
Program Jacobi_Local_Barrier;

Const 

  n = 4;

  numiter = 5;

Var 

  i, j : Integer;

  a, b : Array [0..n+1,0..n+1] of Real; 

  higher, lower : Array [1..n] of Channel of Integer;

Procedure LocalBarrier (i: Integer);

var 


  dummy : Integer;

begin

  If i > 1 then dummy:=lower[i];  (*Αποστολή στη διεργασία I-1*)

  If i < n then begin

    dummy := higher[i];           (*Λήψη από τη διεργασία I-1*)

    lower[i+1] := 1;              (*Αποστολή στη διεργασία I+1*)

  end;

  If i > 1 then higher[i-1] := 1; (*Λήψη από τη διεργασία I-1*)

end;

Begin
(*Κυρίως πρόγραμμα*)

 (*Αρχικοποίηση των τιμών για τον πίνακα Α*)

. . . . . . 

  B := A;

  Forall i := 1 to n do (@i-1)

    Var  
      j, k : Integer;

    begin

      For k := 1 to numiter do begin

        For j := 1 to n do
          B[i,j] := (A[i-1,j]+A[i+1,j]+A[i,j-1]+A[i,j+1])/4;

        LocalBarrier(i);

        A[i] := B[i];

        LocalBarrier(i);

      end;

    end;

 (*Εμφάνιση αποτελεσμάτων*)

. . . . . .
End.
Άσκηση 6η 

Το πρόγραμμα της 5ης άσκησης χρησιμοποιεί μια τεχνική τοπικού φράγματος για συγχρονισμό. Έστω οι ακόλουθες τροποποιήσεις στη διαδικασία τοπικού φράγματος LocalBarrier:

Procedure LocalBarrier(i: Integer);

Var
  dummy: Integer;

begin
  If i > 1 then begin
    higher[i-1]:= 1;

    dummy:= lower[i];

  end;

  If i < n then begin
    lower[i+1] := 1;

    dummy := higher[i];

  end;

end;

Αυτή η νέα διαδικασία επιτυγχάνει τον επιθυμητό τοπικό συγχρονισμό. Όμως, πιέζει όλα τα τοπικά φράγματα να εκτελεστούν ακολουθιακά, ένα κάθε φορά, γεγονός που καταργεί την λειτουργικότητα του τοπικού φράγματος. Εξηγείστε τον λόγο που η διαδικασία LocalBarrier προκαλεί την ακολουθιακή εκτέλεση των τοπικών φραγμάτων και ως αποτέλεσμα παρουσιάζει χρόνο εκτέλεσης O(n) για το συγχρονισμό όλων των n διεργασιών. Χρησιμοποιήσατε κάποια γραφήματα για να διαπιστώσετε και να εξηγήσετε την εμφάνιση του αδιέξοδου; Αν ναι, ποια γραφήματα χρησιμοποιήσατε και πως αυτά σας βοήθησαν;
	1.
	Άσκηση: 
 Πρόγραμμα: 


	2.
	Ώρα έναρξης: 
 Ώρα λήξης: 


	3.
	Εντολές του καταλόγου επιλογών του προγραμματιστικού περιβάλλοντος της MPWin που χρησιμοποιήσατε για την άσκηση (π.χ. File – Load Charts, Parameters – Time Alarm, Charts – Process Activity Chart κοκ): 


	4.
	Προηγήθηκε παρουσίαση της συγκεκριμένης χρήσης της εντολής στην αίθουσα;
α) Εντολή: 
:
( Ναι
( Όχι
β) Εντολή:
:
( Ναι
( Όχι

γ) Εντολή:
:
( Ναι
( Όχι

	5.
	Πόση ώρα χρειαστήκατε για να κατανοήσετε την χρήση-σύνταξη της κάθε εντολής; 


	6.
	Πόση ώρα χρειαστήκατε για την εφαρμογή και την ερμηνεία των αποτελεσμάτων της κάθε εντολής; 


	7.
	Το σύστημα βοήθειας της MPWin όσον αφορά την χρήση της εντολής:
α) 
  ήταν:
( Πλήρες
( Επαρκές
( Μέτριο
( Ελλιπές
β) 
  ήταν:
( Πλήρες
( Επαρκές
( Μέτριο
( Ελλιπές

γ) 
  ήταν:
( Πλήρες
( Επαρκές
( Μέτριο
( Ελλιπές

	8.
	Δώστε μια σύντομη περιγραφή του λόγου χρήσης της κάθε εντολής.




	9.
	Το συγκεκριμένο εργαλείο-εντολή-γράφημα σας βοήθησε (όπου κρίνετε απαραίτητο, σημειώνετε περισσότερες από 1 επιλογές για κάθε εντολή ξεχωριστά):

· στην καλύτερη κατανόηση και αποσαφήνιση των παράλληλων και κατανεμημένων εννοιών

· στην ορθή σύνταξη του κώδικα

· στην γρήγορη και αποτελεσματική αποσφαλμάτωση (debugging)

· στην επεξεργασία των αποτελεσμάτων

· στην εκτίμηση της απόδοσης του παράλληλου προγράμματος.

	10.
	Σημειώστε τυχόν προβλήματα-ελλείψεις που συναντήσατε κατά την χρήση του περιβάλλοντος (είτε στο περιβάλλον είτε στο σύστημα βοήθειας) στην συγκεκριμένη άσκηση:




	11.
	Προτείνετε τρόπους βελτίωσης του περιβάλλοντος ή/και του συστήματος βοήθειας της MPWin:

